Lab 3 – Operational Amplifiers: Part II

Objectives

- To execute the instructions of Lab 3 as provided for ECE 3410 from Canvas (https://usu.instructure.com/).
  - Measurement of non-ideal op amp characteristics such as DC open-loop gain, slew rate, full- power bandwidth, and input offset voltage.

Preparation

Component and Materials

- 741 operational amplifiers (1)
- 100Ω resistor (1)
- 10kΩ resistor (2)
- 100kΩ resistor (1)
- Datasheet for the uA741 operational amplifier

Equipment

- Banana Cable Sets (5)
- Oscilloscope Probes (2)
- Potentiometer Adjustment Tool (1)
- BNC-to-BNC Cable (1)
- BNC-to-Alligator Cable (1)

Pre-Lab Analysis

➤Analytical

Exercise 1

$$\Rightarrow_1 \quad R_1 = 10\text{ k}\Omega \quad R_2 = 10\text{ k}\Omega \quad R_3 = 100\text{ k}\Omega \quad R_4 = 100\ \Omega$$

$$v_{\text{IN}} = 0\text{ V} \quad A = \infty\ (\text{ideal}) \quad I_{\text{bias}} = 0\text{ A} \quad V_X \approx V_{\text{OFS}}$$

$$\Rightarrow_2^{\text{NVA}} \quad \frac{V_{\text{OUT}} - V_Y}{R_2} = \frac{V_Y}{R_1} + \frac{V_{\text{OFS}}}{R_4} \quad I_3 = \frac{V_Y - V_{\text{OUT}}}{R_3} = \frac{V_{\text{OFS}}}{R_4}$$

$$\Rightarrow_3 \quad V_{\text{OUT}} = V_Y\left(1 + \frac{R_2}{R_1}\right) + V_{\text{OFS}}\frac{R_2}{R_4} \quad V_Y = V_{\text{OFS}}\left(1 + \frac{R_3}{R_4}\right)$$

$$\Rightarrow_4 \quad V_{\text{OUT}} = \left[V_{\text{OFS}}\left(1 + \frac{R_3}{R_4}\right)\right]\left(1 + \frac{R_2}{R_1}\right) + V_{\text{OFS}}\frac{R_2}{R_4}$$

$$\Rightarrow_5 \quad V_{\text{OUT}} = V_{\text{OFS}}\left[\left(1 + \frac{R_3}{R_4}\right)\left(1 + \frac{R_2}{R_1}\right) + \frac{R_2}{R_4}\right] = V_{\text{OFS}}G_O$$

$$\Rightarrow_6 \quad \boxed{G_O = 2.102\text{ kV/V}}$$



Figure 1: Circuit for Exercises 1 and 2.

Exercise 2

$\Rightarrow_1$ $R_1 = 10\ \text{k}\Omega \quad R_2 = 10\ \text{k}\Omega \quad R_3 = 100\ \text{k}\Omega \quad R_4 = 100\ \Omega$

$v_{\text{in}} = 1\ \text{V} \quad v_y = 0.01\ \text{V} \quad I_{\text{bias}} = 0\ \text{A}$

$\Rightarrow_2$ $v_y^{(2)} \approx v_{\text{in}}\left(\frac{R_2(R_3+R_4)}{(R_1+R_2)(R_3+R_4)+R_1R_2+AR_1R_4}\right) \quad v_y^{(3)} \approx v_{\text{in}}\left(\frac{R_2(R_3+R_4)}{AR_1R_4}\right)$

$\Rightarrow_3$ $A \approx \frac{v_{\text{in}}}{v_y}\left(\frac{R_2(R_3+R_4)}{R_1R_4}\right) \approx \frac{v_{\text{in}}}{v_y}G_a$ $\quad \Rightarrow_4^{\therefore}$ $\boxed{A \approx 100.1\ \text{kV/V}}$ $\quad \Rightarrow_5^{\therefore}$ $\boxed{\begin{array}{l} v_y^{(2)} \approx 9.79\ \text{mV} \\ v_y^{(3)} \approx 10\ \text{mV} \end{array}}$

$\Rightarrow_6^{\therefore}$ $\boxed{\text{The margin of difference between } v_y^{(2)} \text{ and } v_y^{(3)} \text{ is } 2.1\%.}$

Exercise 3

$\Rightarrow_1$ $\text{SR} = 0.5\ \text{V}/\mu\text{s} \quad V_{\text{R}} = \pm 15\ \text{V} \quad V_p = 2\ \text{V}$

$\Rightarrow_2$ $\text{FPBW} = \frac{\text{SR}}{2\pi|V_{\text{R}}|} \quad V_{\max} = V_{\text{R}}\frac{\text{FPBW}}{f_{\max}} \quad V_{\max} = V_p$

$\Rightarrow_3^{\therefore}$ $\boxed{\text{FPBW} = 5.31\ \text{mHz}}$ $\quad \boxed{f_{\max} = 39.79\ \text{mHz}}$



Figure 2: Circuit for Exercise 3.

➤Calculated

exercise1.sp >>

```
Lab 3, Exercise 1, ECE 3410
***************************
* By Chris Winstead
* Measure offset voltage
***************************

* Include the model file:
.include lab_parts.md

* Define a numerical constant for
* estimating VOFS:
.csparam Go=2102

* Power supplies:
VDD ndd 0 DC 15V
VSS nss 0 DC -15V

* The input voltage sources
Vofs n1 0 DC 5mV

* Resistors
R1 0 ny   10k
R2 ny nout 10k
R3 ny nx   100k
R4 nx 0    100

* Op Amp Model
X1 n1 nx ndd nss nout uA741

* Control Commands:
.control
dc Vofs 0 10m 100u

plot v(nout) $&Go*v(n1)

.endc

.end
```

exercise2.sp >>

```
Lab 3, Exercise 2, ECE 3410
***************************
* By Chris Winstead
***************************

* Include the model file:
.include lab_parts.md

* Define a numerical constant for
* estimating VOFS:
.csparam Ga=100100

* Power supplies:
VDD ndd 0 DC 15V
VSS nss 0 DC -15V

* The input voltage sources
Vin n1 0 DC 0V SIN(0 1 5.0)

* Resistors
R1 n1 ny   10k
R2 ny nout 10k
R3 ny nx   100k
R4 nx 0    100

* Op Amp Model
X1 0 nx ndd nss nout uA741

* Control Commands:
.control

tran 10m 10s
plot v(n1) v(ny)

meas tran VY  PP v(ny) FROM=0 TO=10s
meas tran VIN PP v(n1) FROM=0 TO=10s
let A=VIN/VY*$&Ga
print A

.endc

.end
```

exercise3.sp >>

```
Lab 3, Exercise 3, ECE 3410
***************************
* By Chris Winstead
***************************

* Include the model file:
.include lab_parts.md
.param f=50k
.csparam f=f
.csparam T=2/f

* Power supplies:
VDD ndd 0 DC 15V
VSS nss 0 DC -15V

* The input voltage sources
Vin n1 0 DC 0V SIN(0 2 f)

* Op Amp Model
X1 n1 nout ndd nss nout uA741

* Control Commands:
.control

tran 10n 1m
linearize
wrdata slewing v(nout)

fourier $&f v(nout)

plot xlimit 0 $&T v(n1) v(nout)

fft v(nout)
plot xlog xlimit 1000 10e6 vdb(nout)
wrdata fft vdb(nout)

.endc

.end
```

log.txt >>

exercise1.sp >>

```
In the plot produced, it is
observed that V_OUT and G_o*V_OFS
are identical until the v-sweep
reaches 7.083 mV. After this
point, V_OUT saturates at 14.194
V while G_o*V_OFS continues along
the line of linear voltage
increase. This behavior is
expected and correlates correctly
with the pre-lab analysis.
```

exercise2.sp >>

```
Frequency, f | Gain (Open-Loop), A

0.5 Hz | 197362 V/V
5.0 Hz | 139560 V/V

It is observed that the open-loop gain
at 0.5 Hz and 5 Hz differs by 34.3%.
This is reasonably close to the
projected 30% difference described in
the lab.
```

exercise3.sp >>

```
f_max = 39.79 mHz
(Pre-Lab - Exercise 3)

Frequency, f | THD
| maxslope (V/s) | minslope (V/s)

30 kHz | 0.166% | 381530 | -376570
40 kHz | 1.116% | 489760 | -487540
50 kHz | 6.546% | 502600 | -499410

Slew-Rate (50 kHz), SR = 0.5026 V/µs

**See Plots E.3 30K, 40K, 50K (FFT)
in Appendix.
** See Plot E.3 50K (T) in Appendix.
```

Lab Experiments

Procedure 1

Resistance, Measured (R_1) = 9.83 kΩ (1.7% err.)
Resistance, Measured (R_2) = 9.97 kΩ (0.3% err.)
Resistance, Measured (R_3) = 98.4 kΩ (1.6% err.)
Resistance, Measured (R_4) = 98 Ω (2.0% err.)

Gain, Experimental (G_o) = 2.126 kV/V (1.1% err.)
Gain, Experimental (G_a) = 1.019 kV/V (1.8% err.)

Procedure 2

v_{IN} = 0 V V_R = ±15 V

Measured (V_{OUT})	Experimental (V_{OFS})
1.75 V	823.06 μV

Procedure 3

	v_{in}	v_y
Peak-to-Peak, Measured =	23.3 V	8 mV
Zero-to-Peak =	11.65 V	4 mV

Gain (Open-Loop), Experimental (A) = 2.97 MV/V (2867% err.)

Procedure 4

$$SR = \frac{\Delta v_{\text{out}}}{\Delta t}$$

Voltage, Measured (Δv_{out}) = 150 mV
Time, Measured (Δt) = 0.25 μs

Slew Rate (SR) = 0.6 V/μs (19.4% err.)

Procedure 5

At either 40 kHz or 30 kHz…

- No distortion is perceived in the output waveform.
- Only marginal distortion is perceived in the FFT display.

Commentary

- Improved familiarization and increased experience was obtained in closed-loop and open-loop op-amp gain calculations and analysis. Additionally, the effects of slewing in the FFT plot and display were observed and analyzed.
- The results derived from the analytical, calculated, and experimental methods of the lab for exercise 1 and associated procedures 1 and 2 yielded positive correlation and consistency with marginal error. This demonstrates that the closed-loop analysis will yield reliable results. However, the derived value of V_{OFS} was substantially smaller than all other reported values by fellow lab colleagues for this experimental procedure.
- Conversely, while the analytical and calculated results of exercise 2 positively correlated with each other, the experimental analysis in the associated procedure 3 for measuring open-loop gain yielded a substantially larger value on a much greater order of magnitude. Exhaustive attempts were made to find fault in the experimental analysis that would have yielded a more sensible value, but the end conclusion even with the agreement of the lab supervisor was that the value would be accepted. This resulted in an open-loop gain value substantially higher than all other reported values by fellow lab colleagues for this experimental procedure.
- Finally, the results derived from the analytical, calculated, and experimental methods of the lab for exercise 3 and procedures 4 and 5 likewise yielded positive correlation and consistency with marginal error. The measured slew rate derived during the experimental procedure yielded a result that matched the joint average slew rate values produced from other colleagues in the lab session. In procedure 5, it was observed that a frequency of at least 90 kHz was needed before a clearly apparent slewing distortion could be visible.

Appendix

Plot E.3 30K (FFT)



Plot E.3 40K (FFT)



Plot E.3 50K (FFT)



Plot E.3 50K (T)